# $Id: PKGBUILD 184149 2013-05-02 19:39:50Z eric $
# Maintainer:
# Contributor: Sergej Pupykin <pupykin.s+arch@gmail.com>
# Contributor: Jeff Mickey <jeff@archlinux.org>
# Contributor: Michel Brabants <michel.brabants@euphonynet.be>

pkgname=libzrtpcpp
pkgver=2.3.2
pkgrel=1
pkgdesc="A C++ implementation of Phil Zimmermann's ZRTP specification"
arch=('i686' 'x86_64')
url='http://www.gnutelephony.org/index.php/GNU_ZRTP'
license=('GPL3')
depends=('ccrtp')
makedepends=('cmake')
source=(http://ftp.gnu.org/gnu/ccrtp/${pkgname}-${pkgver}.tar.gz{,.sig})
md5sums=('b5f0a72252785862133e9cbc9cbbd3f8'
         'SKIP')

prepare() {
  sed 's/lib64/lib/g' -i ${pkgname}-${pkgver}/CMakeLists.txt
}

build() {
  mkdir build
  cd build
  cmake ../${pkgname}-${pkgver} \
    -DCMAKE_INSTALL_PREFIX=/usr \
    -DCMAKE_BUILD_TYPE=Release
  make
}

package() {
  cd build
  make DESTDIR="${pkgdir}" install
}
